This maple script estimates the magnitude of the elastic strain energy of a nonhydrostatically stressed isotropic solid.

The derivation corrects a math error in the expression of Connolly (Equation 29 of J.A.D. Connolly, 2009, Geochemistry Geophysics Geosystems, 10:10, Q10014, doi:10.1029/2009GC002540) who gives the total strain energy of an isotropic solid as

U = V0 * ((e[1]^2 + e[2]^2 + e[3]^2) * c[1,1] / 2
 + (e[1]*e[2] + e[1]*e[3] + e[2]*e[3]) * 2 * c[1,2]
 + (e[4]^2 + e[5]^2 + e[6]^2) * c[4,4] / 2)

rather than

U = V0 * ((e[1]^2 + e[2]^2 + e[3]^2) * c[1,1] / 2
 + (e[1]*e[2] + e[1]*e[3] + e[2]*e[3]) * c[1,2]
 + (e[4]^2 + e[5]^2 + e[6]^2) * c[4,4] / 2),

the difference being a factor of two in the second term of the second factor, where e[i] and c[i,j] are the elastic strain components and stiffness coefficients. The strain components e[i] are related to the elements of the strain tensor e[j,k] as: e[1] = e[1,1], e[2] = e[2,2], e[3] = e[3,3], e[4] = 2*e[2,3] = 2*e[3,2], e[5] = 2*e[1,3] = 2*e[3,1], e[6] = 2*e[2,1] = 2*e[1,2].

To make an order of magnitude assesment of the effect, non-hydrostatic (deviatoric and differential) stresses are assumed to be of magnitude dsig ("delta sigma" in Equation 30 of Connolly 2009) and the shear modulus (mu) is estimated as one third the bulk modulus (K).

Below

C - stiffness matrix
c[i,j] - stiffness matrix component i,j
Epsilon1 - vector form of the strain tensor in Voigt notation.
Sigma1 - vector form of the stress tensor in Voigt notation.
DU - the total elastic strain energy (relative to the unstrained state).
strain_energy - the nonhydrostatic strain energy

the derivation follows from Callen 1985 assuming a Hookean elastic solid.

```
> restart;
> for i from 1 to 6 do;
> for j from 1 to 6 do;
>  c[i,j] := 0 :
> end do;end do;
> eq := 2*mu *(1 + vnu) = 3* K*(1-2*vnu);
```

```
> nu := solve (eq,vnu);
> E := 2*mu*(1+nu);
```

compliance matrix elements for an isotropic solid:

```
> c11 := 2*mu + (K-2/3*mu); c44 := mu; c12 := c11 - 2*c44;
> c[1,1] := c11: c[2,2] := c[1,1]: c[3,3] := c[1,1]: c[1,2] := c12:
  c[1,3] := c[1,2]: c[2,3] := c[1,2]: c[2,1] := c[1,2]: c[3,1] :=
  c[1,2]: c[3,2] := c[1,2]: c[4,4] := c44: c[5,5] := c[4,4]: c[6,6]
  := c[4,4]:
> with(LinearAlgebra):
> C := Matrix(6,6,c);
> Sigma1 :=
  Vector(6,[sigma[1],sigma[2],sigma[3],sigma[4],sigma[5],sigma[6]]):
> Epsilon1 :=
  Vector(6,[epsilon[1],epsilon[2],epsilon[3],epsilon[4],epsilon[5],e
  psilon[6]]):
> Sigma := collect(simplify(C.Epsilon1),{mu,K});
```

compliance matrix S, Epsilon is the strain components in terms of stress.

```
> S := MatrixInverse(C);
> Epsilon := collect(simplify(S.Sigma1),{mu,K});
```

total strain energy/V0

```
> total_strain_energy := 0:
> for k from 1 to 6 do:
>  total_strain_energy := total_strain_energy +
  Sigma[k]*epsilon[k]/2 :
> end do:
```

$$eq := 2\,\mu\,(1+vnu) = 3\,K\,(1-2\,vnu)$$

$$\nu := \frac{3\,K-2\,\mu}{2\,(3\,K+\mu)}$$

$$E := 2\,\mu\left(1+\frac{3\,K-2\,\mu}{2\,(3\,K+\mu)}\right)$$

$$c11 := \frac{4\,\mu}{3}+K$$

$$c44 := \mu$$

$$c12 := K-\frac{2\,\mu}{3}$$

$$C := \begin{bmatrix} \frac{4\mu}{3}+K & K-\frac{2\mu}{3} & K-\frac{2\mu}{3} & 0 & 0 & 0 \\ K-\frac{2\mu}{3} & \frac{4\mu}{3}+K & K-\frac{2\mu}{3} & 0 & 0 & 0 \\ K-\frac{2\mu}{3} & K-\frac{2\mu}{3} & \frac{4\mu}{3}+K & 0 & 0 & 0 \\ 0 & 0 & 0 & \mu & 0 & 0 \\ 0 & 0 & 0 & 0 & \mu & 0 \\ 0 & 0 & 0 & 0 & 0 & \mu \end{bmatrix}$$

$$\Sigma := \begin{bmatrix} K(\varepsilon_1+\varepsilon_2+\varepsilon_3)+\left(\frac{4}{3}\varepsilon_1-\frac{2}{3}\varepsilon_2-\frac{2}{3}\varepsilon_3\right)\mu \\ K(\varepsilon_1+\varepsilon_2+\varepsilon_3)+\left(-\frac{2}{3}\varepsilon_1+\frac{4}{3}\varepsilon_2-\frac{2}{3}\varepsilon_3\right)\mu \\ K(\varepsilon_1+\varepsilon_2+\varepsilon_3)+\left(-\frac{2}{3}\varepsilon_1-\frac{2}{3}\varepsilon_2+\frac{4}{3}\varepsilon_3\right)\mu \\ \mu\,\varepsilon_4 \\ \mu\,\varepsilon_5 \\ \mu\,\varepsilon_6 \end{bmatrix}$$

$$S := \begin{bmatrix} \frac{3K+\mu}{9\mu K} & -\frac{3K-2\mu}{18\mu K} & -\frac{3K-2\mu}{18\mu K} & 0 & 0 & 0 \\ -\frac{3K-2\mu}{18\mu K} & \frac{3K+\mu}{9\mu K} & -\frac{3K-2\mu}{18\mu K} & 0 & 0 & 0 \\ -\frac{3K-2\mu}{18\mu K} & -\frac{3K-2\mu}{18\mu K} & \frac{3K+\mu}{9\mu K} & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{1}{\mu} & 0 & 0 \\ 0 & 0 & 0 & 0 & \frac{1}{\mu} & 0 \\ 0 & 0 & 0 & 0 & 0 & \frac{1}{\mu} \end{bmatrix}$$

$$E := \begin{bmatrix} \dfrac{\frac{1}{3}\sigma_1 - \frac{1}{6}\sigma_2 - \frac{1}{6}\sigma_3}{\mu} + \dfrac{\frac{1}{9}\sigma_1 + \frac{1}{9}\sigma_2 + \frac{1}{9}\sigma_3}{K} \\ \dfrac{-\frac{1}{6}\sigma_1 + \frac{1}{3}\sigma_2 - \frac{1}{6}\sigma_3}{\mu} + \dfrac{\frac{1}{9}\sigma_1 + \frac{1}{9}\sigma_2 + \frac{1}{9}\sigma_3}{K} \\ \dfrac{-\frac{1}{6}\sigma_1 - \frac{1}{6}\sigma_2 + \frac{1}{3}\sigma_3}{\mu} + \dfrac{\frac{1}{9}\sigma_1 + \frac{1}{9}\sigma_2 + \frac{1}{9}\sigma_3}{K} \\ \dfrac{\sigma_4}{\mu} \\ \dfrac{\sigma_5}{\mu} \\ \dfrac{\sigma_6}{\mu} \end{bmatrix}$$

the bulk strain is tr(epsilon)/3*I, drop I

```
> Eps_tr := (epsilon[1] + epsilon[2] + epsilon[3] )/3;
```

eps_tr is the symbolic version of Eps_tr/3:

```
> dilational_strain_energy :=
  collect(subs(epsilon[1]=eps_tr,epsilon[2]=eps_tr,epsilon[3]=eps_tr
  ,epsilon[4]=0,epsilon[5]=0,epsilon[6]=0,total_strain_energy),{mu,K
  });
> deviatoric_strain_energy := collect(simplify(subs(epsilon[2] =
  -(epsilon[1]+epsilon[3]),subs(eps_tr=Eps_tr,total_strain_energy -
  dilational_strain_energy))),{mu,K});
```

$$Eps\_tr := \frac{1}{3}\varepsilon_1 + \frac{1}{3}\varepsilon_2 + \frac{1}{3}\varepsilon_3$$

$$dilational\_strain\_energy := \frac{9\,K\,eps\_tr^2}{2}$$

$$deviatoric\_strain\_energy := \mu\left(2\,\varepsilon_1^{\,2} + 2\,\varepsilon_1\,\varepsilon_3 + 2\,\varepsilon_3^{\,2} + \frac{1}{2}\varepsilon_4^{\,2} + \frac{1}{2}\varepsilon_5^{\,2} + \frac{1}{2}\varepsilon_6^{\,2}\right)$$

```
> simplify(deviatoric_strain_energy);
```

$$\frac{1}{2}\mu\,(4\,\varepsilon_1^{\,2} + 4\,\varepsilon_1\,\varepsilon_3 + 4\,\varepsilon_3^{\,2} + \varepsilon_4^{\,2} + \varepsilon_5^{\,2} + \varepsilon_6^{\,2})$$

```
> deviatoric_strain_energy_in_deviatoric_strain :=
  collect(simplify(subs(epsilon[1]=epsilon1[1]+eps_tr,epsilon[3]=eps
  ilon1[3]+eps_tr,epsilon[2]=-(epsilon1[1]+epsilon1[3])+eps_tr,devia
  toric_strain_energy)),{mu,K});
```

deviatoric_strain_energy_in_deviatoric_strain :=

$$\left(2\,\varepsilon 1_1^{\,2}+2\,\varepsilon 1_1\,\varepsilon 1_3+2\,\varepsilon 1_3^{\,2}+\frac{1}{2}\,\varepsilon_4^{\,2}+\frac{1}{2}\,\varepsilon_5^{\,2}+\frac{1}{2}\,\varepsilon_6^{\,2}\right)\mu$$

```
> deviatoric_strain_energy_in_stress :=
  collect(subs(epsilon[1]=Epsilon[1],epsilon[2]=Epsilon[2],epsilon[3
  ]=Epsilon[3],epsilon[4]=Epsilon[4],epsilon[5]=Epsilon[5],epsilon[6
  ]=Epsilon[6],deviatoric_strain_energy),{mu,K});
```

$$\begin{aligned}
\mathit{deviatoric_strain_energy_in_stress} := {} & \left(\frac{4}{3}\left(\frac{1}{3}\,\sigma_1-\frac{1}{6}\,\sigma_2-\frac{1}{6}\,\sigma_3\right)^2+\frac{2}{3}\left(-\frac{1}{6}\,\sigma_1+\frac{1}{3}\,\sigma_2-\frac{1}{6}\,\sigma_3\right)^2\right.\\
& -\frac{2}{3}\left(-\frac{1}{6}\,\sigma_1+\frac{1}{3}\,\sigma_2-\frac{1}{6}\,\sigma_3\right)\left(-\frac{1}{6}\,\sigma_1-\frac{1}{6}\,\sigma_2+\frac{1}{3}\,\sigma_3\right)+\frac{2}{3}\left(-\frac{1}{6}\,\sigma_1-\frac{1}{6}\,\sigma_2+\frac{1}{3}\,\sigma_3\right)^2+\frac{1}{2}\,\sigma_4^{\,2}+\frac{1}{2}\,\sigma_5^{\,2}\\
& \left.+\frac{1}{2}\,\sigma_6^{\,2}\right)\Big/\,\mu+\left(2\left(\frac{1}{3}\,\sigma_1-\frac{1}{6}\,\sigma_2-\frac{1}{6}\,\sigma_3\right)\left(\frac{1}{9}\,\sigma_1+\frac{1}{9}\,\sigma_2+\frac{1}{9}\,\sigma_3\right)\right.\\
& +\frac{2}{3}\left(-\frac{2}{9}\,\sigma_1-\frac{2}{9}\,\sigma_2-\frac{2}{9}\,\sigma_3\right)\left(\frac{1}{3}\,\sigma_1-\frac{1}{6}\,\sigma_2-\frac{1}{6}\,\sigma_3\right)+\frac{2}{3}\left(\frac{1}{9}\,\sigma_1+\frac{1}{9}\,\sigma_2+\frac{1}{9}\,\sigma_3\right)\left(-\frac{1}{6}\,\sigma_1+\frac{1}{3}\,\sigma_2-\frac{1}{6}\,\sigma_3\right)\\
& \left.+\frac{2}{3}\left(\frac{1}{9}\,\sigma_1+\frac{1}{9}\,\sigma_2+\frac{1}{9}\,\sigma_3\right)\left(-\frac{1}{6}\,\sigma_1-\frac{1}{6}\,\sigma_2+\frac{1}{3}\,\sigma_3\right)\right)\Big/\,K\\
& +\frac{\left(\frac{4}{3}\left(\frac{1}{9}\,\sigma_1+\frac{1}{9}\,\sigma_2+\frac{1}{9}\,\sigma_3\right)^2+\frac{2}{3}\left(-\frac{2}{9}\,\sigma_1-\frac{2}{9}\,\sigma_2-\frac{2}{9}\,\sigma_3\right)\left(\frac{1}{9}\,\sigma_1+\frac{1}{9}\,\sigma_2+\frac{1}{9}\,\sigma_3\right)\right)\mu}{K^2}
\end{aligned}$$

sigma1[i] are the deviatoric principle stresses, sig_tr is the mean stress

```
> deviatoric_strain_energy_in_deviatoric_stress := V0 *
  collect(simplify(subs(sigma[1]=sigma1[1]+sig_tr,sigma[3]=sigma1[3]
  +sig_tr,sigma[2]=-(sigma1[1]+sigma1[3])+sig_tr,deviatoric_strain_e
  nergy_in_stress)),{dsig,mu,K});
```

to evakuate the magnitude of the deviatoric strain energy assume all independent deviatoric stresses are ~ dsig

```
> approximate_deviatoric_strain_energy_in_deviatoric_stress := V0 *
  collect(simplify(subs(sigma[4]=dsig,sigma[5]=dsig,sigma[6]=dsig,si
  gma[1]=dsig+sig_tr,sigma[3]=dsig+sig_tr,sigma[2]=-(dsig+dsig)+sig_
  tr,deviatoric_strain_energy_in_stress)),{dsig,mu,K});
```

$$deviatoric\_strain\_energy\_in\_deviatoric\_stress := \frac{V0\left(\frac{1}{2}\sigma 1_1^{\ 2}+\frac{1}{2}\sigma 1_1\,\sigma 1_3+\frac{1}{2}\sigma 1_3^{\ 2}+\frac{1}{2}\sigma_4^{\ 2}+\frac{1}{2}\sigma_5^{\ 2}+\frac{1}{2}\sigma_6^{\ 2}\right)}{\mu}$$

$$approximate\_deviatoric\_strain\_energy\_in\_deviatoric\_stress := \frac{3\,V0\,dsig^2}{\mu}$$

```
> bulk_strain_energy_in_mean_stress := V0*
  collect(simplify(subs(sigma[1]=sigma1[1]+P,sigma[3]=sigma1[3]+P,si
  gma[2]=-(sigma1[1]+sigma1[3])+P,subs(epsilon[1]=Epsilon[1],epsilon
  [2]=Epsilon[2],epsilon[3]=Epsilon[3],subs(eps_tr=Eps_tr,dilational
  _strain_energy)))),{mu,K});
```

$$bulk\_strain\_energy\_in\_mean\_stress := \frac{V0\,P^2}{2\,K}$$

observing that mu ~ K/3, then the non-hydrostatic strain energy is comparable to the bulk strain energy when:

```
> eq :=
  subs(mu=K/3,approximate_deviatoric_strain_energy_in_deviatoric_str
  ess) = bulk_strain_energy_in_mean_stress;
```

$$eq := \frac{9\,V0\,dsig^2}{K} = \frac{V0\,P^2}{2\,K}$$

solving this equation for dsig gives the magnitude of the deviatoric/differential stress at which the deviatoric strain energy is comparable to the volumetric strain energy in terms of pressure.

```
> solve(eq,dsig);
```

$$\frac{\sqrt{2}\,P}{6}, -\frac{\sqrt{2}\,P}{6}$$

as the bulk strain energy grows as P^2 the deviatoric strain energy becomes insignificant at pressures > sqrt(18)*dsig ~ 4 dsig.